

CASO PRÁCTICO PRÉSTAMOS. CÁLCULO DE CUOTAS

Nuestra empresa necesita adquirir una nueva nave industrial por la que debe pagar 700.000,00 €
Para financiar el pago solicitaremos un préstamo hipotecario a nuestro banco. Las condiciones del préstamo son las siguientes:

- El importe del préstamo ascenderá al 80 % del valor de la nave.
- El tipo de interés aplicado será del 1,75% nominal anual.
- El periodo de amortización será de 25 años.

TRABAJO A REALIZAR:

1.- Calcular el importe de cada cuota si realizamos pagos mensuales.

2.- Calcular la tabla de amortización del préstamo para el primer año de vigencia del mismo diferenciando entre intereses y amortización del principal.

SOL. CASO PRÁCTICO PRÉSTAMOS. CÁLCULO CUOTAS

Para resolver este ejercicio necesitamos conocer las fórmulas financieras que se utilizan en los cálculos de los préstamos de tipo francés o, en su defecto, las funciones financieras que utiliza Excel para resolver este tipo de préstamos.

Excel incorpora varias funciones que permiten efectuar cálculos financieros sin necesidad de construir modelos complicados o tener que utilizar las clásicas fórmulas financieras.

Aunque en este CASO solamente necesitamos conocer la función que calcula el pago de las cuotas de un préstamo, vamos a presentar toda la serie de funciones que sirven para calcular los distintos elementos asociados con un préstamo de tipo francés, como por ejemplo el número de periodos en que se amortiza el préstamo, o el tipo de interés nominal aplicado, entre otros.

Funciones para calcular un préstamo

Las funciones financieras que incorpora Excel para el cálculo de los distintos elementos de un préstamo, utilizan una serie de argumentos comunes a todas ellas que definen los distintos elementos asociados a dicha operación financiera. Estos argumentos son:

- **TASA:** es el tipo de interés nominal por periodo.
- **NPER:** es el número total de periodos de pago.
- **PAGO:** es el importe que se paga en cada periodo. Contiene una parte correspondiente al principal y otra a intereses.
- **VA:** es el valor actual que se percibe (o se desembolsa) al principio de la operación. Se introduce en las fórmulas con signo negativo cuando se trata de un préstamo.
- **VF:** es el valor final o valor futuro que se percibe (o se desembolsa) al final de la operación.
- **TIPO:** indica si la operación es postpagable, en cuyo caso Tipo=0, o si es prepagable, en cuyo caso Tipo=1.

Además, estos argumentos siguen siempre el mismo orden dentro de las funciones, que es precisamente el que hemos seguido para su definición.

Para operar con estas funciones hay que conocer una serie de cuestiones básicas:

- **TASA, NPER** y **PAGO** deben hacer siempre referencia al mismo periodo de tiempo, es decir, si vamos a calcular los pagos mensuales, deberemos referir Nper a meses y Tasa deberá ser el tipo de interés mensual; si los pagos son trimestrales, Nper estará referido a trimestres y Tasa deberá ser el tipo de interés trimestral, etc...
- **VF** y **TIPO** suelen ser 0 en todos los préstamos, por lo que se pueden omitir de la función. Sin embargo, habrá que conservarlos si **TIPO = 1**, ya que habrá que respetar su posición dentro de la función.
- La **TASA** se calcula como: Tipo de interés nominal anual / Periodicidad.
- Si analizamos un préstamo, **VA** debe considerarse un valor negativo, dado que hace referencia al importe que debemos.
- Estas funciones se aplican también a operaciones de constitución de capital con cuota constante.

1.- En el primer apartado, tenemos que calcular el pago mensual a realizar para amortizar un préstamo hipotecario. La función que utilizaremos será la siguiente:


PAGO:

=PAGO(Tasa;Nper;Va;Vf;Tipo)


Comenzaremos introduciendo en la hoja de cálculo los datos necesarios para calcular el pago a realizar. No debemos introducir datos en las celdas de cálculos, ya que en ese caso, las fórmulas, deberán modificarse cada vez que deseemos hacer un cambio en los datos. Por tanto, el modelo contendrá una celda para cada dato que vayamos a utilizar en el cálculo de la cuota. Dado que se trata de una operación con pagos mensuales, habrá que transformar los años en meses y calcular la tasa de interés mensual para lo que precisaremos conocer además la periodicidad de la operación:

	A	B	C
1			
2			
3		Valor de la nave industrial	700.000,00 €
4		% de Hipoteca	80%
5		Importe de la hipoteca (VA)	560.000,00 €
6		Tipo de interés anual	1,75%
7		interés del periodo (TASA)	0,15%
8		Periodo de amortización (en años)	25
9		Número de periodos (NPER)	300
10		PERIODICIDAD:	12
11			

Para obtener la fórmula de la función PAGO, tenemos dos alternativas, teclearla, o bien acudir al menú FÓRMULAS/BIBLIOTECAS DE FUNCIONES/INSERTAR FUNCIÓN, con lo que aparecerá la pantalla correspondiente al Asistente para funciones donde seleccionaremos la categoría de funciones financieras y dentro de éstas la función PAGO:



Podemos ver que Excel muestra los argumentos, con lo cual si no los recordamos los tenemos aquí. Al pulsar “Aceptar” pasamos a la segunda pantalla del asistente para funciones, desde el cual nos irá pidiendo las celdas para cada uno de los argumentos proporcionando información sobre cada uno de ellos:



A pesar de que la función PAGO tiene cinco argumentos, podemos observar que en este caso los dos últimos son cero, dado que el valor final de un préstamo siempre será cero y si el modelo es postpagable el argumento Tipo=0. Tendremos en cuenta además que el argumento correspondiente al Valor actual se pondrá con signo negativo, dado que el valor actual representa lo que tenemos en este momento y en el caso de un préstamo lo que tenemos es una deuda por el importe concedido. La función puede ser cualquiera de las siguientes:

=PAGO(C7;C9;-C5;0;0)

=PAGO(C7;C9;-C5;;)

=PAGO(C7;C9;-C5)

Dado que son equivalentes elegiremos la tercera ya que es la más breve.

En caso de que el modelo fuese prepagable, las dos opciones válidas serían las siguientes:

=PAGO(C7;C9;-C5;0;1)

=PAGO(C7;C9;-C5;;1)

El modelo completo en fórmulas quedará así:

	A	B	C
1			
2			
3		Valor de la nave industrial	700000
4		% de Hipoteca	0,8
5		Importe de la hipoteca (VA)	=C3*C4
6		Tipo de interés anual	0,0175
7		interés del periodo (TASA)	=C6/C10
8		Periodo de amortización (en años)	25
9		Número de periodos (NPER)	=C8*C10
10		PERIODICIDAD:	12
11			
12			
13		PAGO	=PAGO(C7;C9;-C5)
14			

El pago mensual será:

	A	B	C
1			
2			
3		Valor de la nave industrial	700.000,00 €
4		% de Hipoteca	80%
5		Importe de la hipoteca (VA)	560.000,00 €
6		Tipo de interés anual	1,75%
7		interés del periodo (TASA)	0,15%
8		Periodo de amortización (en años)	25
9		Número de periodos (NPER)	300
10		PERIODICIDAD:	12
11			
12			
13		PAGO	2.306,02 €
14			

2.- En este caso vamos a utilizar dos nuevas funciones muy similares a la función PAGO, las funciones PAGOPRIN y PAGOINT.

PAGOPRIN y PAGOINT:

Estas funciones son una variante de la función PAGO, cuyos argumentos son los mismos que en el caso anterior, pero incluyen, antes del argumento Nper, el argumento Per que hace relación al periodo concreto de que se trata, en el primer periodo Per=1, en el segundo periodo Per=2, etc.

=PAGOPRIN(Tasa;Per;Nper;Va;Vf;Tipo)

=PAGOINT(Tasa;Per;Nper;Va;Vf;Tipo)

Para construir la tabla de amortización empezaremos poniendo en una columna los números correspondientes a los periodos, dado que se trata de pagos mensuales el primer año estará compuesto por doce periodos, con lo cual, en la columna de periodos (en nuestro caso será la columna B) tendremos los números 1 a 12. A continuación, en la celda C15 escribiremos la función PAGO, en la celda D15 la función PAGOPRIN y en la E15 la función PAGOINT:

=PAGO(C7;C9;-C5)

=PAGOPRIN(C7;B18;C9;-C5)

=PAGOINT(C7;B18;C9;-C5)

Los valores de las funciones para los restantes periodos, los obtendremos llenando hacia abajo con AUTOLLENADO, pero hay que analizar cuidadosamente que es lo que va a variar de unos periodos a otros y que es lo que va a permanecer constante. Así, **Tasa, Nper y Va**, coincidirán en todos los periodos, lo que quiere decir que las funciones deben hacer referencia a las mismas celdas en todos ellos. Por el contrario el argumento **Per** es variable por propia definición. Por tanto, en la función PAGO, todos los argumentos deberán presentar referencias absolutas antes de llenar hacia abajo, mientras que en las funciones PAGOPRIN y PAGOINT, el argumento **Per** deberá mantenerse como referencia relativa de modo que al llenar hacia abajo, vaya cambiando la referencia y calculándose en función del periodo de que se trate.

Las funciones quedarán:

=PAGO(\$C\$7;\$C\$9;-\$C\$5)

=PAGOPRIN(\$C\$7;B18;\$C\$9;-\$C\$5)

=PAGOINT(\$C\$7;B18;\$C\$9;-\$C\$5)

La tabla de amortización resultante será la siguiente:

	A	B	C	D	E
16					
17					
18					
19					
20					
21					
22					
23					
24					
25					
26					
27					
28					
29					
30					
31					

	Periodo	Cuota	Amortización	Intereses
18	1	2.306,02 €	1.489,35 €	816,67 €
19	2	2.306,02 €	1.491,52 €	814,49 €
20	3	2.306,02 €	1.493,70 €	812,32 €
21	4	2.306,02 €	1.495,88 €	810,14 €
22	5	2.306,02 €	1.498,06 €	807,96 €
23	6	2.306,02 €	1.500,24 €	805,78 €
24	7	2.306,02 €	1.502,43 €	803,59 €
25	8	2.306,02 €	1.504,62 €	801,40 €
26	9	2.306,02 €	1.506,82 €	799,20 €
27	10	2.306,02 €	1.509,01 €	797,00 €
28	11	2.306,02 €	1.511,22 €	794,80 €
29	12	2.306,02 €	1.513,42 €	792,60 €

Podemos comprobar que –tal como hemos construido el modelo- podemos calcular el préstamo con otras condiciones de tipo de interés, número de años de amortización, etc., sin más que cambiar los datos correspondientes.

CASO PRÁCTICO PRÉSTAMOS. CÁLCULO DE CUOTAS. Ampliación

3.- Supongamos ahora que podemos modificar el periodo de amortización del préstamo entre 20 y 30 años. Realizar un análisis de sensibilidad para determinar los importes de la cuota mensual que resultarían en cada caso.

4.- Realizar un nuevo análisis de sensibilidad considerando la variación anterior en el número de años y teniendo también en cuenta la posibilidad de que, simultáneamente, el tipo de interés oscile entre el 1,25 % y el 2,25 % en intervalos de un cuarto de punto porcentual.

5.- Realizar los cambios oportunos en la TABLA anterior para que se muestren únicamente los importes de la cuota en el caso de que éstos no superen los 2.400 € -límite que no deseamos sobrepasar en los pagos mensuales- en caso de que se supere dicho límite, las celdas correspondientes deberán aparecer en blanco.

**Las herramientas para solucionar estos apartados están explicadas en
detalle en la solución al Caso:
*CASO PRÁCTICO PRÉSTAMOS. CÁLCULO DE PERIODOS Y VA***

